
News

US Training Days: Thursday 11  th   and Friday 12  th   October  

These training days will take place at the DoubleTree by Hilton New Orleans, 300 
Canal Street, New Orleans just prior to the Society for Neuroscience meeting in New 
Orleans. The days are suitable for both existing and prospective users so join us and 
learn how to make the most of Spike2 and Signal for your individual data acquisition 
and analysis requirements.

Attendance is limited so please register early to reserve your place on the appropriate 
course.

Future meetings and events

2012 UCL Neuroscience Symposium
University College London,
London, UK
June 29th 2012

Physiology 2012
Edinburgh International Conference Centre,
Edinburgh, UK
July 2nd – 5th 2012

8th FENS Forum of Neuroscience 2012
International Convention Center (CCIB),
Barcelona, Spain
July 14th – 18th 2012

ISEK 2012
Brisbane Convention and Exhibition Centre,
Brisbane, Australia
July 19th – 21st 2012

International Motoneuron Meeting 2012
Crowne Plaza Hotel Coogee,
Sydney, Australia
July 23rd – 26th 2012

Society for Neuroscience 2012
Ernest N Morial Convention Center,
New Orleans, LA, USA
October 13th - 17th 2012

Latest versions of Spike2 and Signal

Updates for Spike2 and Signal are available from the CED Downloads page, or by 
clicking on the links in the table below. Demonstration versions of the latest software 
are also available.

Spike2 downloads Released Signal downloads Released

Spike2 version 7.10 05/12 Signal version 5.06 05/12

Spike2 version 6.16 03/11 Signal version 4.09 08/11

Spike2 demo 05/12 Signal demo 05/12

Scripts Spotlight

There are times, as a script writer, that you will want to re-use some of your user-
defined procedures and functions in multiple script projects. Of course, you can do this 
by simply copying and pasting the code but it can be much more convenient to use the 
#include command, which is available in Spike2 version 6.03 and higher. This 
command lets you access procedures and functions in other script files from within 
your current project. Simply use the following syntax in your script:

#include “filename”

where filename is either an absolute pathname  (“C:\scripts\MyFuncs.s2”) or a 
filename enclosed with angle brackets (<MyFuncs.s2s>). If you only provide a 
filename the script expects the script file you want to include to either reside in the 
same folder as the original script, in the current folder or in the Spike\include 
folder. It is usually a good idea to have all of your #include commands at the start of 
your script so that anyone else who reads it understands that other script files are 
being used.

Many of the scripts in your Spike\scripts folder make use of #include to read 
functions from a script file called GHutils.s2s. This file contains a number of useful 
functions that I utilise on a regular basis and can be found in your Spike\include 
folder. Another useful script called seqlib.s2s can also be found here. This script 
contains a library of functions to help calculate sequencer variable values.

Q. I have recently started using Spike2 for spike detection and found the video in 
the previous eNewsletter very helpful for it's description of the do's and don'ts 
of setting trigger levels. Now I’m looking at how my templates are being 
created and the settings I can use to define them. Can you tell me more about 
the Template parameter settings and how they work?

A. There are a number of parameters used to define how spike templates are 
formed and how incoming spikes are matched to templates. These parameters 
can be modified using the Template Parameters dialog which is available from 
a toolbar button in the spike shape window as indicated below.

Spike shape window

The parameters dialog includes four areas that provide settings for creation of 
new templates, matching incoming spike data to templates, template 
modification and controls for processing the raw waveform data into spikes.

Template parameters

New template

This section controls the creation of new templates. The Number of spikes for 
a new template field is used to set the number of spikes required for a 
provisional template to be promoted to a confirmed template. The initial 
template width is determined by the New template width as a percentage of 
amplitude field. As spikes are added to a template, the template width 
changes to represent the amplitude variation of the spikes. 

The No template for shapes rarer than 1 in N spikes can be used to ignore 
provisional templates that are created from threshold crossings that do not 
match any spike shapes in n number of detected spikes. This can prevent 
templates being created for random noise or artefacts that cross the spike 
trigger thresholds. It also decrements the provisional template counts if no 

spikes are added within the specified number, allowing the provisional 
templates under consideration to be kept to the minimum amount.

Matching a spike to a template

The fields in this section determine how incoming spikes are matched to 
existing templates. Using the Maximum percent amplitude change for a match 
field, Spike2 can scale every incoming spike up or down by this value so that 
the area under the spike matches the area under the target template. This 
setting can be very useful if you have spikes that maintain their shape but 
change amplitude during recording.

The Minimum percentage of points in template field sets the percentage of the 
spike data points that must fit in the template to confirm a match. The final 
option can be used to specify that the Minimum percentage of points... setting 
is only used when building templates. If this option is checked then the number 
of points in the template are ignored and spikes are matched to the template 
with the smallest error between the mean template and the spike.

Template maintenance

This group determines how the shape of a template changes over time as 
more spikes are added to it. The Template modification mode setting has three 
options available from the drop-down list. Add All will add all spikes that match 
the template, modifying the template accordingly to represent the amplitude 
variation of each spike. Auto Fix lets you fix, or 'lock' the template once a set 
number of spikes has been added. If you have several similar spikes, using 
this mode to lock templates after a small number of spikes can stop a template 
gradually changing shape and becoming the same as another template. Track 
mode sets the template shape to track the incoming spikes. This mode is only 
useful for slow changes in spike shapes as, with fast changes, it is likely that 
templates will merge together.

Waveform data

This final group of settings controls how the raw waveform data is processed 
for spike sorting. Waveform interpolation method sets how the spike 
waveforms are aligned with templates using linear, parabolic or cubic spline 
interpolation. For on-line sorting, linear interpolation is always used for 
efficiency.

If your spike waveforms show baseline drift you can use the High-pass filter 
time constant to remove slow changes in the baseline. If your waveform shows 
fast transient baseline changes a Remove the DC offset option is also 
available. 

Scripts: Spike2

Q. Is it possible to detect peaks in a result view? I would like to generate a PSTH 
and then mark the first peak above mean baseline + two standard deviations 
after the trigger time.

A. The script, PSTH peak detect.s2s, allows the user to generate a PSTH 
and then automatically detects the first peak after the trigger time larger than 
baseline mean plus a number of Standard Deviations. It then measures peak 
amplitude and latency and prints these values to the Log window. You can set 
how many consecutive bins must be above the threshold to define a peak in 
the settings dialog generated by the script. 

plot by selecting the 
appropriate FrameVar value 
in the Trend Plot settings 
dialog

'$PSTHPeakDetect.s2s|Script to create a PSTH and automatically place curors, find peaks, measure peak amplitude and latency
'You can set the number of consecutive bins that are required to define a peak and level below threshold
'Creates a PSTH with minimum 0.5 s baseline before time 0 and assumes that peak will follow time 0
'sets excitation threshold at mean + 2SD

'The script is a 'work in progress' and is offered without guarantees.
'You must test it to see whether it is suitable for your application.

'CED 07/06/12

var PSTH%; 'View created by SetPsth
var dataFile%;
Var Chan%:=1; 'Channel to analyse
Var SDMul:=2;
Var ok%;

Var BINS%:=1; 'Number of consecutive bins to detect above
var ch%,trigCh%;
var width:= 1.0;
var bSize:=0.01;
var offset:=0.5;
var noBins%;

if ViewKind() = 0 then              'Check we have a data file open
dataFile%:= View();
FrontView(dataFile%);
else
dataFile%:= FileOpen("",0);        'If not prompt user to select one
WindowVisible(1);
endif;

if dataFile% < 0 then
    Message("Could not open a data file!");     'Quit if no file opened
    halt;endif;
View(LogHandle());
EditSelectAll();
EditClear();
View(dataFile%);

repeat
DlgCreate("PSTH settings");
DlgChan(1,"Channel",26);
DlgReal(2,"Width (s)",0.01,1000);
DlgReal(3,"Bin Size (s)",0.001,1000);
DlgReal(4,"Offset (s)",0.5,1000);
DlgChan(5,"Trigger", 26);
    ok%:= DlgShow(ch%,width,bSize,offset,trigCh%);
    if ok% = 0 then
       Halt;endif; 

noBins%:= width/bSize;
if noBins% = 0 then
    Message("Cannot create PSTH with 0 bins!");
endif;
until noBins% > 0;
 
PSTH%:= SetPsth(ch%,noBins%,bSize,offset,trigCh%);
WindowVisible(1);                      'Make it visible
Process(0.0, MaxTime(), 0, 1);          'Analyse
Window(0,0,50,50);                     'Size the window

If ViewKind(PSTH%)<> 4 then                 'check that active window is a STH
    Message("Wrong view! Halting");
    Halt
endif

DlgCreate("Setup");  'Start new dialog
DlgChan(1,"Channel to analyse",1);
DlgInteger(2,"Number of consecutive bins for peak",1,100,0,0);
DlgReal(3,"Number of standard deviations above mean baseline",1.000000,10.000000,0,0);
ok% := DlgShow(chan%, Bins%, SDMul);    'ok% is 0 if user cancels,  variables updated if not

if ok% <0 then
        Message("Cancelled! Halting");
    Halt
endif

CursorPlace%();
FrontView(LogHandle());
WindowVisible(1);

Func CursorPlace%();
Var Mean;
Var SD;

var Level;
var Level2;
var i%;
Var ct%;
Var c2%;
Var h1%;
Var h2%;
var h3%;
Var Pos;
Var Value;
Var binsz;


binsz:=BinSize(chan%); 'Get histogram bin width
binsz:=binsz/2;  'Divide by two to place cursor mid bin

HCursorDelete(-1);                       'delete any existing cursors
CursorDelete(-1);

CursorSet(3);                             'place cursors to define baseline
Cursor(1,XToBin(-0.5));                   'change -0.5 to other value if need different duration baseline
Cursor(2,XToBin(0));
CursorLabel(4,1,"Start base");
CursorLabel(4,2,"End Base");
CursorLabel(4,3,"Peak");
'CursorLabel(4,4,"End");


mean:=ChanMeasure(Chan%,2, BinToX(Cursor(1)),  BinToX(Cursor(2)));        'calculate baseline mean
SD:=ChanMeasure(Chan%,12, BinToX(Cursor(1)),  BinToX(Cursor(2)));          'calculate SD

Level:=Mean+(SD*SDMul);                                                     'define value of + threshold, change "2" if need other setting
Level2:=Mean-SD;                                                         'define value of - threshold

PrintLog("%s\n",Windowtitle$());
PrintLog("Bins for significance\t%.3f\n",BINS%);                         'write values to log file
PrintLog("Threshold (+2SD)\t%.3f\n",Level);
'PrintLog("Threshold(-SD)\t%.3f\n",Level2);
PrintLog("Baseline Mean\t%.3f\n",Mean);


H1%:=HCursorNew(1,Level);                                                  'display horizontal cursor at + threshold
HCursorLabel(4,H1%,Print$("T(+SD*%.1f)",SDMul));                                           'label it

'if Level2 >0 then;                                                         'check if - threshold is > 0, if so, place cursor
'    H3%:=HCursorNew(1,Level2);
'    HCursorLabel(4,H3%,Print$("T(-SD*1)"));
'    endif

For i%:= XToBin(0) to XToBin(Maxtime()) do                                 'find bin after time 0 > threshold
     
Pos:=BinToX(i%)+binsz;
Cursor(3,XToBin(Pos));    
Value:=ChanValue(Chan%,Pos);    
    if Value > Level then;
        Ct%:=Ct%+1;
         'Cursor(3,i%);                                                                  'check if consecutive bins (# set by value of bins% variable) above threshold
        if ct%=BINS% then;
            Cursor(3,i%-(BINS%-1));                                         'if so, place vertical cursor at beginning of 1st bin of peak
            i%:=XToBin(Maxtime());
        endif            
    else
        ct%:=0;        
    endif
    Next  

if ct%=BINS% then 
    'find bin falls below threshold
    For i%:=Cursor(3)+BINS% to XToBin(Maxtime()) do
        
        Pos:=BinToX(i%)+binsz;
        'Cursor(4,XtoBin(pos));
        Value:=ChanValue(Chan%,Pos);
        
'    if Value < Level then;
' 	C2%:=C2%+1;
        
'        if c2%=BINSBelow% then;                                                  'check for consecutive bins
'            Cursor(4,I%+1);                                         'if so place cursor'
'		i%:=XToBin(Maxtime());
'	endif
 '   else
 '       c2%:=0;        
 '   endif
    Next 
 
        Mean:=ChanMeasure(Chan%,2, BinToX(Cursor(3)),  BinToX(Cursor(4))+1);          'calculate mean in bin
        H2%:=HCursorNew(cHan%, Mean);                                                'place cursor at peak mean
        HCursorLabel(4,H2%,"Peak mean");
            i%:=XToBin(Maxtime());
            PrintLog("Peak Mean\t%.3f\n",Mean);                                       'wtite values to log
PrintLog("Latency\t%.3f\n",BinToX(Cursor(3)));
'PrintLog("Duration\t%.3f\n\n",BinToX(Cursor(4))-BinToX(Cursor(3)));
        
else
PrintLog("Failed to detect peak");    
endif


'Interact("Place cursors", 1023);

return 1
end


CED
File Attachment
PSTH Peak detect.s2s



Q. Is it possible to set-up different channels in a data file so that they are 
displayed in different colours?

A. Signal supports individual colour settings for data, channels and views. Click 
on the colour palette icon in the main toolbar or select Change colours… 
from the View menu to access the Set colours dialog. 

Set colours dialog

With a data file open, the Set colours dialog will display a list of the channels 
in the file. The drop-down menu can then be used to select the different 
channel elements such as primary, secondary and background colour for the 
listed channels. Click on the name of the channel you want to change and then 
click a colour in the palette to apply it to the selected channel element.

            Set custom colours

The available colours in the palette can be changed by double clicking on a 
colour. This opens a new dialog where you can select Basic colours to add to 
the palette or define your own Custom colours with user-definable settings for 

Back to top

Back to top

Right-click the script icon 
and save to disk.

If you have any problems 
opening the embedded 
scripts in this newsletter 
please let us know

Back to top

Right-click the script icon 
and save to disk.

If you have any problems 
opening the embedded 
scripts in this newsletter 
please let us know

Back to top

Back to top

Right-click the sequence 
icon and save to disk.

If you have any problems 
opening the embedded files 
in this newsletter please let 
us know

Back to top

CED
Hint
The default colours used in Signal can be changed by selecting Application colours in the drop-down menu of the Set colours dialog.



Hue, Saturation, Luminescence or RGB ratio. 

Channel colour settings are stored in the resource file that is associated with 
the data file. Resource files have the same name as the data file and are 
stored in the same location with the extension .sgr. You can apply colour 
settings from one file to other data files using the File menu Resource 
Files > Apply Resource Files… option. 

Scripts: Signal

Q. Can I set a data file to only display frames of a given state? For example, I 
want to review all frames of state 1 while paging through a file, without having 
to scroll through frames of all other states.

A. The script, StateDisplay.sgs, allows you to set a state to display and then 
page through the file using the associated script toolbar buttons to only show 
frames of the selected state.

Did you know...?

You can record events or triggers as digital markers in Signal using the checkbox 
option in the General tab of the sampling configuration. You could use this to log the 
times of a pulse from a stimulator, for example. Digital markers are read from the 
digital input port on the rear of the 1401.

Recent Questions

Q. Is there anyway to log data from additional devices (for example, a motion 
tracking system) alongside the data I am recording with Spike2 and my 1401?

A. Spike2 supports the use of Talker interfaces which can be used to read data 
from external devices. Talkers are custom software modules that act as an 
interface between 3rd party data capture devices and Spike2. The output from 
supported devices can be added as additional channels to the same file being 
recorded using a 1401 interface. As an example, consider a motion tracking 
system that outputs co-ordinate values to a PC via a suitable connection, for 
example, an ethernet connection. This output could be added automatically to 
a sampling Spike2 file by writing a Talker interface.

Four example Talkers are included with Spike2. To install these, along with
documentation describing how to write your own custom modules, select the
Custom setup option during installation of Spike2 and check the 
Talker documentation option. Channels from Talker modules are added to the 
sampling configuration from the drop-down menu next to the New button.

'$StateDisplay|Script to only display frames of a selected state number

'The script is a 'work in progress' and is offered without guarantees.
'You must test it to see whether it is suitable for your application.

'CED 10/05/12

var dataFile%;		'Handle of data file
var numFrames%;		'Number of frames
var state%;         'Selected state
var stateNo$;       'Text display for toolbar
var index%;         'Index counter for current frame
var maxFrames%;     'Max. no. of frames of state n
var ok%;            ' Check for dialog entry    
var dispFrames%[1000];  'Array to hold all frames
var maxState%;      'Current state count
var cFrame%;        'No. of current frame
var cFrame$;        'Text of current frame no.
var name$;
var states%[256];

dataFile% := FrontView();		'Show an open data file

if ViewKind(dataFile%) <> 0 then
	Message("No Signal data file open\nPlease select a data file");
	dataFile%:= FileOpen("",0,1);	'if no data file found, prompt user to open one
	
	if dataFile% < 0 then				'If user cancels or error occurs
   		Message("Unable to open a file!\nRun script to try again");		'Warn user 		
		Halt;		'and quit script	
	endif;
endif;
name$:= FileName$(3);
Window(0,0,100,100);	'Make the data full screen

numFrames%:= FrameCount();		'Get no. of frames
GetFrameStates%(dispFrames%[]);		'Get frame states
DoToolbar%();


Func GetFrameStates%(dispFrames%[]);
var c%;                 'Counter for frames
var d%:=1;              'Counter for states
var currState%;         'State of current frame
var s%;

for c%:=1 to numFrames% do		'Cycle through frames
	
	currState%:= FrameState(c%);	'get current state
    
    if states%[currState%] = 0 then
        states%[currState%]:= 1
        endif;
	
	if currState% > maxState% then
		maxState%:= currState%;		'If state number greater than previous, set to max state
	endif;
next;

	DlgCreate("State Filter");
	DlgInteger(1,"State to display",0,maxState%);		'User selects state to display
	ok%:= DlgShow(state%);

	if ok% < 0 then
		Message("Script cancelled");
	Halt;endif;

	stateNo$:= Print$("Display of State %d",state%);	'Toolbar display
    
for c%:=1 to numFrames% do      'Cycle through frames
	if FrameState(c%) = state% then     'And match state to state filter
		dispFrames%[d%]:= c%;
		d%+=1;                  'Keep count of frames that match
	endif;
next;

if d%=1 then
    View(LogHandle());
    EditSelectAll();
    EditClear();
    PrintLog("States available in %s: ",name$);
    for s%:=1 to maxState%+1 do
        if states%[s%-1] = 1 then
            PrintLog("%d,",s%-1);
        endif;
    next;
    
    FrontView(LogHandle());
    Window(0,0,50,20);
    Message("This file does not contain any frames of State %d\n A list of available States for this file can be found in the Log window",state%);
    Halt;endif;

dispFrames%[0]:= d%-1;
maxFrames%:= d%-1;            'Get count of frames that match
Frame(dispFrames%[1]);      'Display the first frame of state n
cFrame%:= Frame();          'Get current frame no.
cFrame$:= Print$(": Frame %d",cFrame%);     'Print for toolbar display
index%:= 1;
return 1;
end;

Func DoToolbar%()           'Toolbar to allow control of frame display
ToolbarClear();
ToolbarSet(1,"&Quit");      
ToolbarSet(3,">>",Fwd%);
ToolbarSet(4,"<<",Rew%);
return Toolbar(stateNo$+cFrame$,1023);  'Display current frame and state
end;

Func Fwd%()
if index% < maxFrames% then;        'If current display is not the last frame in the list
	index% += 1;                    'Increment index
	Frame(dispFrames%[index%]);     'Display the next frame
    cFrame%:= Frame();              'Get current frame no.
    cFrame$:= Print$(": Frame %d",cFrame%);     
    ToolbarText(stateNo$+cFrame$);      'Update the toolbar display
endif;
return 1;
end;

Func Rew%()
if index% > 1 then                  'If current display is not the first frame in the list
	index% -= 1;                    'Decrement index
	Frame(dispFrames%[index%]);     'Display the previous frame
    cFrame%:= Frame();              'Get current frame no.
    cFrame$:= Print$(": Frame %d",cFrame%);
    ToolbarText(stateNo$+cFrame$);      'Update the toolbar display
endif;
return 1;
end;


CED
File Attachment

Sampling configuration showing installed Talker interfaces

We happily provide the information to enable manufacturers of other data 
capture devices to produce their own Talker modules, but do feel free to 
contact us if you have a particular device that you would like to interface with 
Spike2.

User group
Try the CED Forums bulletin board for software and hardware support

If you have any comments about the newsletter format and content, or wish to 
unsubscribe from the mailing list for this newsletter, please notify sales@ced.co.uk.

